
Lab 8: Modeling Bivariate Longitudinal Data

Overview

Don’t use one longitudinal variable as a time-varying predictor of another longitudinal variable.  Using it as a predictor usually oversimplifies the interrelationship, often massively.  Instead, a bivariate longitudinal model should be used, where covariates like treatment and coping style are used to predict both longitudinal variables.  We then assess the residual correlation between the longitudinal variables.  
1. Create an appropriate data set

2. Bivariate random intercept 

a. Random intercepts correlated?

b. Residual errors correlated?

c. Three special cases.
3. Unstructured covariance model versus independent model

4. Product Correlation Models

5. Bivariate random intercept and slope

6. Factor Analytic Models

Use the Pediatric Pain data for this lab.  Pain rating is a 1-10 rating of how painful the subject found each cold pressor trial.  There is one rating and one pain tolerance measure for each trial.  We consider the bivariate longitudinal relationship between pain tolerance and pain rating.  

There are substantial amounts of fixed effects that are output.  We will be ignoring those for the most part.  If we wanted to inspect the fixed effects, we generally would be required to include a substantial number of estimate statements to get interpretable output.  
Lab Tasks

We will be inspecting the fit statistics and covariance parameters.  We will want to do LRTs between nested models, and we will look at the covariance parameters to see what the correlation is between pain tolerance and pain rating.  

TODO:  (1)  Keep track in a table (by hand or by SAS or otherwise) of the different covariance models, the number of covariance parameters, the REML log likelihood, AIC and BIC.  

TODO: (2) Always check for convergence of the models.  

TODO: (3) At home, make sure you understand the parameterization of the covariance matrix.  

1.  Create an appropriate data set

There are 4 time points.  The data set needs to be constructed with 8 observations per subject in the variable response, and another variable, type, is created that indicates whether the response is a pain rating (type = r) or a pain tolerance (type = t) variable.  

* Restructure dataset to long format for bivariate analysis;
data pain;

*infile desktop\pain.txt';
infile 'f:\class\m236\pain.txt'; /*CHANGE THIS PATH IF PAIN.TXT NOT IN THIS DIRECTORY*/
input id  ses  age_years age_months gender tmt0 cs0 helpful 



easy tol11 tol12 rat11 rat12 tol13 tol14 rat13 rat14 senscat1 cpantic1 ;

*CHANGE MISSING CODE TO MISSING VALUES;
  array v(10) age_months ses rat11 rat12 rat13 rat14 tol11 tol12 tol13 tol14;

  do i=1 to 10;



if v(i)=-999 then v(i)=.;


end;

** RECODE To bivariate long format from wide.  ;
response=log2(tol11); type="t"; trial=1; cs = cs0; copingstyle="A"; 



if cs0=2 then copingstyle="D"; tmt=0; treatment="Z"; output;

response=log2(tol12); type="t"; trial=2; output;

response=log2(tol13); type="t"; trial=3; output;

response=log2(tol14); type="t"; trial=4; 
tmt=tmt0;  



treatment="A"; if tmt0=2 then treatment="D"; if tmt0=3 then treatment="N"; output;

response=rat11; type="r"; trial=1; tmt=0; treatment="Z"; output;

response=rat12; type="r"; trial=2; output;

response=rat13; type="r"; trial=3; output;

response=rat14; type="r"; trial=4; tmt=tmt0; treatment="A"; 

  

if tmt0=2 then treatment="D"; if tmt0=3 then treatment="N"; output;

drop ses  age_years age_months gender helpful easy tol11 tol12 rat11 rat12 tol13 tol14 

    
rat13 rat14 senscat1 cpantic1 i tmt0 cs0;

run;

proc contents;

run;

proc print;

run;

TODO:  (4) (a) Inspect the data set enough to understand that there are 8 observations per subject and how the type, copingstyle and treatment variables are coded.  (b) Understand the code above.  

TODO: (5) How many observations?  WHY?  

2. Bivariate random intercept

******************

* 

* The Bivariate Random intercept model.  

* 

******************;
proc sort data=pain;

by id trial type;

run;

title "1 Corr RI + Corr res";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


random type/ type=un subject=id g gcorr;


repeated / type=un subject=trial(id) r rcorr;

run;

TODO (6) Remember: Check for convergence of the model.  

TODO (7) How many missing observations?  

TODO (8) Look at the covariance parameters.  Is the correlation between the random intercepts significant?  Is the correlation between the rating and tolerance residuals significant?  

3. The three models nested in the random intercept model
title "2 Corr RI + Ind res";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


random type/ type=un subject=id g gcorr;


repeated / type=un(1) subject=trial(id) r rcorr;

run;

title "3 Ind RI + Corr res";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


random type/ type=un(1) subject=id g gcorr;


repeated/ type=un subject=trial(id) r rcorr;

run;

title "4 Ind RI + Ind res";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


random type/ type=un(1) subject=id g gcorr;


repeated / type=un(1) subject=trial(id) r rcorr;

run;

Todo (9)  Remember to be filling out your LL/AIC/BIC table.  

Todo (10)  Which of these four models fit best?  Is there a significant correlation of some type between rating and tolerance in this model?  

4. Unstructured covariance model versus independent model
********************************

* 

* UN covariance structure, two different parameterizations.  

* Should give the same parameter estimates and the same log likelihoods.  

* Easier to understand the covariance parameters if we sort by type first.  

* 

********************************;
proc sort data=pain;


by id type trial;


run;

title "5 completely UN";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated/ type=un subject=id r rcorr;

run;

title "6 completely UNR";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated/ type=unr subject=id r rcorr;

run;
********************************

* 

* Rating & Tolerance are Independent
* Separate UN covariance matrices for each variable.  

* 

********************************;
proc sort data=pain;


by type id trial;


run;

title "7 UN Ind";

proc mixed method=reml covtest data=pain;


by type;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated/ type=unr subject=id r rcorr;

run;

TODO:  (11) What are the the correlations within rating?  Within tolerance?  Between rating and tolerance?  

TODO:  (12) Continue filling out the table.  
TODO:  (13) Add the REML log likelihoods from these two models to get the reml log likelihood from the uncorrelated model.  Compare this value to the UN REML log likelihood.  Is the correlated model significantly better than the uncorrelated model?  This test is generally of low power, because it uses so many parameters for fitting the correlations between the two variables.  

5. Product correlation models

******************

* 

* Product correlation models.  

* 

******************;
proc sort data=pain;


by id trial type;


run;

title "9 UN@UN trial type";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated trial type/ type=un@un subject=id r rcorr;

run;

title "10 UN@CS trial type";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated trial type/ type=un@cs subject=id r rcorr;

run;

title "11 UN@CS type trial";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated type trial/ type=un@cs subject=id r rcorr;

run;
TODO:  (14) What is the difference between the last two models?  

TODO:  (15) Which of these 3 models fits best?  One of these easily fits the worst (which is it?), so drop that from further consideration.  In the best fitting model, is there a correlation between pain rating and pain tolerance?  

TODO:  (16) Do one of these models seem to be the best of all models fit to this data so far?  Which is it?  What is its interpretation?  

6. Bivariate random intercept and slope
********************************

* 

* Bivariate Random Intercept and Slope Models

* 

********************************;
data pain;

set pain;

time=trial;

run;

title "12 Corr RIAS + Corr res";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


random type type*time/ type=un subject=id g gcorr;

 
repeated/ type=un subject=trial(id) r rcorr;

run;
TODO: (17) Does this data set need a bivariate RIAS model?  Does this model fit better than the bivariate RI model?  

Some data sets require this model, and it works well for them (for example, various Kenya outcomes).

7. Factor analytic models

********************************

* 

* Factor Analytic Models with 1 and 2 factors.  

* 

********************************;
proc sort data=pain;


by id type trial;


run;

title "13 FA1";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated/ type=FA(1) subject=id r rcorr;

run;

title "14 FA2";

proc mixed method=reml covtest data=pain;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


repeated/ type=FA(2) subject=id r rcorr;

run;
TODO:  (18) Which factor analytic model is best?  Are either competitive with the other models?  

/*******************************MACRO FOR ALL MODELS IN A TABLE*****************************************/
data pain;

set pain;

time=trial;

run;

*CREATE MACRO FOR ALL MODELING;
%macro bigmac(i,title, sortorder,by, rantp, randomtype, reptp, repeatedtype, subj);

proc sort data=pain;

by &sortorder;

run;

title "&title";

proc mixed method=reml covtest data=pain;


&by by type;


class id type copingstyle treatment trial;


model response=type*copingstyle type*treatment type*copingstyle*treatment /  noint s;


 &rantp/ type=&randomtype subject=id g gcorr;


 &reptp/ type=&repeatedtype subject=&subj r rcorr;

ods output convergencestatus=cs&i; /*ADDED*/
ods output covparms=c&i fitstatistics=f&i dimensions=d&i;

run;

data new&i (keep=model type descr value);

format descr $30.;

set cs&i(rename=(reason=descr)) c&i(rename=(covparm=descr estimate=value)) f&i d&i ;

if (index(descr,'Res')>0 or index(descr, 'm')>0) and index(descr,'CC')=0 and index(descr,'mm')=0;

type="&title";

model=&i;

run;

%mend bigmac;

%bigmac(1,Corr RI + Corr res, id trial type,*,random type,un,repeated, un, trial(id));

%bigmac(2,Corr RI + Ind res, id trial type,*,random type,un,repeated, un(1),trial(id)); 

%bigmac(3,Ind RI + corr res, id trial type,*,random type,un(1),repeated, un,trial(id)); 

%bigmac(4,Ind RI + Ind res, id trial type,*,random type,un(1),repeated, un(1),trial(id)); 

%bigmac(5,completely UN, id type trial,*,*,,repeated, un,id); 

%bigmac(6,completely UNR, id type trial,*,*,,repeated, unr,id);

%bigmac(7,UN Ind,type id trial,,*,,repeated, un,id); 

%bigmac(9,UN@UN 1 trial type,id trial type,*,*,,repeated trial type, un@un,id); 

%bigmac(10,UN@CS 1 trial type,id trial type,*,*,,repeated trial type, un@cs,id);

%bigmac(11,UN@CS 2 trial type,id trial type,*,*,,repeated type trial , un@cs,id);

%bigmac(12,Corr RIAS + Corr res,id trial type,*,random type type*time,un,repeated,un,trial(id));

%bigmac(13,FA1,id type trial,*,*,,repeated,FA(1),id);

%bigmac(14,FA2,id type trial,*,*,,repeated,FA(2),id);

*SEPARATE DATA SETS BY TYPE;
proc sort data=new7; by descr;run;

data new7 new8;set new7(drop=type); by descr; 

format type $20.;

if first.descr then do;type='UN Ind rating'; output new7; end;else do; type='UN Ind tolerance'; model=8; output new8; end;

run;

*MACRO TO COMBINE data sets;
%macro nprme;

%do i = 1 %to 14;

 new&i

 %end;

 %mend nprme;

 *MACRO TO SORT data sets;
%macro sprme;

%do i = 1 %to 14;

proc sort data=new&i;by type descr;run;

 %end;

 %mend sprme;

 %sprme;

 data newtotal;

 format type $40.;

 set %nprme;

 by type;

 if first.type then count=1;

 else count+1;

run;

proc print data=newtotal;run;

proc freq data=newtotal;tables count;run;

proc sort data=newtotal;by model type count;run;

proc transpose data=newtotal out=wide(drop= _NAME_ value6) prefix=value;

by model type;

var value;

id count;

run;

* MACRO TO COPY convergence status;

%macro gomac;

%do i=1 %to 14;

data _null_;

set newtotal;

if model=&i and count=6;

call symput('con', descr);

run;

data wide; set wide;

if _n_ = &i then value="&con";

run;

%end;

%mend;

%gomac;

data wide1;

set wide;

by model;

*Manually fix number of parameters;
if model in (2,3,9) then value7=value7-1;

if model = 4 then value7=value7-2;

if model<7 then output;

*Combine independent model types;
if model in (7,8) then do;output;model+lag(model)*10; value7+lag(value7); value1+lag(value1);

value2+lag(value2); value3+lag(value3);value4+lag(value4); type="UN Ind";

end;

if model>8 then output;

*Calculate differences;
if model=1 then do; call symput('v1',value1);call symput('df1',value7);end;

if model=4 then do; call symput('v4',value1);call
symput('df4',value7);end;
if model=5 then do; call symput('v5',value1);call symput('df5',value7);end;

data wide2;set wide1;

if model<5 or model=12 then do vs1=round(abs(value1-&v1),.01);p1=1-probchi(vs1,abs(value7-&df1));end;

if model<5 then do vs4=round(&v4-value1,.01);p4=1-probchi(vs4,value7-&df4);end;

if model not in(5,7,8) then do;vs5=round(value1-&v5,.01);df=&df5-value7;p=1-probchi(vs5,df);end;

run;

options pagesize=150 linesize=120;

proc print data=wide2 label;

title 'Final Table - All 14 models';

label
value1='REML: -2 Res Log Likelihood'
value2='AIC (smaller is better)'
value3='BIC (smaller is better)'
value4='Columns in X'
value5='Columns in Z per Subject'
value='Converge';

value7='Number of Parameters';

var model type value4 value7 value1 value2 value3 vs1 p1 vs4 p4 vs5 df p value;

format type $20. value4 value5 value7 2. p p1 p4 3.2;

run;

